



LUND UNIVERSITY

Simulation Programs for the Grodsky Insulin-Glucose Model

Hagander, Per

1977

Document Version:

Publisher's PDF, also known as Version of record

[Link to publication](#)

Citation for published version (APA):

Hagander, P. (1977). *Simulation Programs for the Grodsky Insulin-Glucose Model*. (Technical Reports TFRT-7124). Department of Automatic Control, Lund Institute of Technology (LTH).

Total number of authors:

1

General rights

Unless other specific re-use rights are stated the following general rights apply:

Copyright and moral rights for the publications made accessible in the public portal are retained by the authors and/or other copyright owners and it is a condition of accessing publications that users recognise and abide by the legal requirements associated with these rights.

- Users may download and print one copy of any publication from the public portal for the purpose of private study or research.
- You may not further distribute the material or use it for any profit-making activity or commercial gain
- You may freely distribute the URL identifying the publication in the public portal

Read more about Creative commons licenses: <https://creativecommons.org/licenses/>

Take down policy

If you believe that this document breaches copyright please contact us providing details, and we will remove access to the work immediately and investigate your claim.

LUND UNIVERSITY

PO Box 117
221 00 Lund
+46 46-222 00 00

SIMULATION PROGRAMS FOR THE GRODSKY
INSULIN-GLUCOSE MODEL

PER HAGANDER

Institutionen för Reglerteknik
Lunds Tekniska Högskola
November 1977

Dokumentutgivare
 0670 Lund Institute of Technology
 Handläggare Dept of Automatic Control
 0670
 Författare
 0870 Per Hagander

Dokumentnamn
 0670 REPORT
 Utgivningsdatum
 0674 Nov. 1977

Dokumentbeteckning
 LUTFD2/(TRT-7124)/1-10/(1977)
 Arendsbezeichnung
 0676

1074

Dokumenttitel och undertitel
 1870
 Simulation Programs for the Grodsky Insulin-Glucose Model.

Referat (sammandrag)
 0656
 A separate paper describes the Grodsky packet storage model and its application to the intravenous glucose tolerance test (IVGTT). This report contains SIMNON systems for the developed models. It also contains the commands and parameters used for the generation of the diagrams of the paper. The MACRO-facility and PARAMETER-SAVE files are utilized.

Referat skrivet av
 0670
 Author

Förslag till ytterligare nyckelord
 0670

Klassifikationssystem och -klass(er)
 0670

Indextermer (ange källa)
 0670

Omfång 107 pages	Övriga bibliografiska uppgifter 5672
Språk English	

Sekretessuppgifter 0670	ISSN 0674	ISBN 0676
----------------------------	--------------	--------------

Dokumentet kan erhållas från 0670 Department of Automatic Control Lund Institute of Technology Box 725, S-220 07 LUND 7 SWEDEN	Mottagarens uppgifter 0674
--	-------------------------------

Pris
 0670

DOKUMENTATABLAD enligt SIS 62 10 12

SIS-DB 1

09-NOV-77
DIRECTORY LISTING (SCR)

1246 FREE BLKS

13 USER FILES

62 USER BLKS

TWOST	SRC	11	09-NOV-77
IVGTT	SRC	2	09-NOV-77
GLUC	SRC	3	09-NOV-77
SDT20	SRC	6	09-NOV-77
INSUL	SRC	11	09-NOV-77
FIG5	SRC	2	09-NOV-77
FIG6	SRC	2	09-NOV-77
ART1	SRC	3	09-NOV-77
ART2	SRC	3	09-NOV-77
ART3	SRC	3	09-NOV-77
VEN1	SRC	10	09-NOV-77

CONTINUOUS SYSTEM TWOST

```

"
" PER HAGANDER 770505
" GRODSKY J CLIN INVEST VOL 51, 2047, (1972)
" LICKO AND SILVERS, MATH BIOSCI, 27, 319-332(1975)
" PULSE AND STEP GLUCOSE INPUT
"
TIME T
STATE X0 X1 X2 P
DER DX0 DX1 DX2 DP
"
INITIAL
S0=EXP(-K*LN(G0/C))
A0=1/(1+S0)
S1=EXP(-K*LN(G1/C))
A1=1/(1+S1)-A0
A2=1-A0-A1
X0=H0/M
X1=A1/A0*H0/M*(1+M/F)
X2=A2/A0*H0/M*(1+M/F)
H1=BET*(G1-G0)+H0
P=H0
"
OUTPUT
GS=IF T<T1 THEN G0 ELSE IF T<T2 THEN G1 ELSE G0
GS1=IF T<T3 THEN GS ELSE G1
G=GS1-EPS
SR0=M*X0
SRG=IF G>G0 THEN M*X1 ELSE 0           " MICRO U/ML,MIN
SRL=IF T<T1 THEN 0 ELSE M*X1
SR1=IF IND<0.5 THEN SRG ELSE SRL
SR=SR0+SR1                             " MICRO U/ML,MIN
H=IF G>G0 THEN H1 ELSE H0              " MICRO U/ML,MIN
XT=X0+X1+X2                             " MICRO U/ML
DYNAMICS
DP=-ALF*(P-H)
DX0=-SR0+A0*P+F*A0*XT-F*X0
DX1=-SR1+A1*P+F*A1*XT-F*X1
DX2=      +A2*P+F*A2*XT-F*X2
"
"PARAMETERS
IND:0           "0 - GRODSKY      1 - LICKO IE ALL REPLENISHMENT IS RELEASED
T1:7           "MIN              TIME OF FIRST STEP UP
T2:40          "                 TIME OF STEP DOWN
T3:50          "                 TIME OF SECOND STEP UP
EPS:0.1
C:120          "MG/100 ML
K:3.3
H0:0.2         "MICRO U/ML,MIN
BET:0.01       "0.1 MU/MG,MIN
F:0.3          "1/MIN
ALF:0.1        "1/MIN
M:0.5          "1/MIN
"
G0:75          "MG/100 ML      BASAL LEVEL
G1:250         "MG/100 ML      HIGH LEVEL
END

```

CONNECTING SYSTEM IVGTT

```

"MODEL 1
TIME T
GI[GLUC]=IF T<TI THEN DOS/TI ELSE 0
GI[INSUL]=GI[GLUC]
DERG[INSUL]=DERG[GLUC]
TI:2           "MIN
DOS:25        "G
END

```

CONTINUOUS SYSTEM GLUC

```

"MODEL 1      OCT 1976
"TWO EXPONENTIAL IMPULSE RESPONSE
"DERIVATIVE OUTPUT INCLUDED
INPUT GI           "G/MIN
OUTPUT DERG G
STATE X1 X2
DER DX1 DX2
"
DX1=-A1*X1+B1*GI/V
DX2=-A2*X2+(1-B1)*GI/V
DERG=DX1+DX2
G=X1+X2           "MG/100 ML
"
A1:0.016         "1/MIN
A2:0.614         "1/MIN
B1:0.233
V:0.0532        "100 LITER
END

```

SDT20

CONTINUOUS SYSTEM DATA

"PATIENT #20

"ARTERIAL DATA

OUTPUT G DAT I DAT

TIME T

"

S=T-EPST

GS=IF S<T1 THEN G1 ELSE IF S<T2 THEN G2 ELSE IF S<T3 THEN G3 ELSE G4

GS2=IF S<T4 THEN GS ELSE IF S<T5 THEN G5 ELSE IF S<T6 THEN G6 ELSE G7

GDAT=IF S<T7 THEN GS2 ELSE G8

IS=IF S<T1 THEN I1 ELSE IF S<T2 THEN I2 ELSE IF S<T3 THEN I3 ELSE I4

IS2=IF S<T4 THEN IS ELSE IF S<T5 THEN I5 ELSE IF S<T6 THEN I6 ELSE I7

IDAT=IF S<T7 THEN IS2 ELSE I8

T1:2 "MIN

T2:4

T3:12

T4:20

T5:30

T6:50

T7:60

G1:67 "MG/100 ML

G2:382

G3:232

G4:161

G5:146

G6:135

G7:119

G8:109

I1:8 "MICRO UNITS/ML

I2:154

I3:109

I4:40

I5:33

I6:19

I7:18

I8:15

EPST:0.1

END

```

CONTINUOUS SYSTEM INSUL
" PER HAGANDER MAY 1977
" INSULIN SECRETION MODEL
" DERIVATIVE INPUT REQUIRED
" TWO POSSIBILITIES FOR INITIAL THRESHHOLD DISTRIBUTIONS.
"
INPUT G DERG
STATE I Y2 X ACC
DER DI DY2 DX DACC
"
OUTPUT
"DIFFERENT POSSIBILITIES FOR THE GAMMA FUNCTION
GTOK=(CRK/(G+G0))*K
GAM0=1/(1+GTOK)
GAM1=IF G<0 THEN 0 ELSE IF G<G1 THEN G/G1 ELSE 1
GAM=IF LG<0.5 THEN GAM0 ELSE GAM1
DGAM0=K*GTOK/(G+G0)/(1+GTOK)/(1+GTOK)
DGAM1=IF G<0 OR G>G1 THEN 0 ELSE 1/G1
DGAM=IF LG<0.5 THEN DGAM0 ELSE DGAM1
DG=MAX(0,DERG)
"
" Y2 - REPLENISHMENT      Y1 - RELEASE OF INSULIN STORED IN LABILE POOL
" Y - INSULIN RELEASE
Y1=DG*DGAM*(XMAX+(IF ISR<0.5 THEN 0 ELSE ACC))
Y=Y1+Y2*(IF ISR<0.5 THEN 1 ELSE GAM)
" SRD=1: SECRETION-DISTRIBUTION DYNAMICS INCLUDED, 1/M MIN.
SR=IF SRD<0.5 THEN Y ELSE X
"
DYNAMICS
DY2=ALF*(BET*G-Y2)
DACC=Y2
DX=-M*(X-Y)
"
DI=-N*I+SR
"
" PARAMETERS      *****
"
CRK:200           " MG/100 ML
K:3.3
G0:67            " MG/100 ML
G1:300           " MG/100 ML
LG:1             " 1 - MEANS LINEARIZED GAMMA FUNCTION
ISR:0            " 0 - MEANS LICKO IE ALL REPLENISHMENT IS RELEASED
SRD:0            " 0 - MEANS NO SECRETION-DISTRIBUTION DYNAMICS
"
M:0              " 1/MIN
XMAX:240         " MICRO U/ML
N:0.33           " 1/MIN
ALF:0.5          " 1/MIN
BET:0.065        " 0.1 MU/MG,MIN
"
GPLOT=GSC*(G+G0)
IPLLOT=ISC*(I+I0)
I0:8             " MICRO U/ML
GSC:1
ISC:1
"
END

```



```
MACRO FIG5
PLOT GPLOT IPLOT GDAT IDAT
AXES
GET ART1
WRITE 'FIG 5'
SIMU 0 61
DISP A1 A2 B1 V
DISP ALF BET XMAX N
DISP ISR
PLOT IPLOT
GET ART2
SIMU 0 61 1
DISP ALF BET XMAX N
GET ART3
SIMU 0 61
DISP ISR
DISP ALF BET XMAX N
END
```

```
MACRO FIG6
PLOT GPLOT IPLOT GDAT IDAT
AXES
WRITE 'FIG6'
GET VEN1
SIMU 0 61 0.1
DISP A1 A2 B1 V
DISP ALF BET XMAX N
DISP SRD M
WRITE '          VENOUS DATA'
END
```

ART1

[INSUL]
I:0.
Y2:0.
X:0.
ACC:0.
CRK:200.
G0:67.
K:3.3
G1:300.
LG:1.
XMAX:240.
ISR:0.
SRD:0.
ALF:0.5
BET:0.065
M:2.
N:0.333
GSC:1.
ISC:1.
IO:8.

ART2

[INSUL]
I:0.
Y2:0.
X:0.
ACC:0.
CRK:200.
G0:67.
K:3.3
G1:300.
LG:1.
XMAX:170.
ISR:0.
SRD:0.
ALF:0.04
BET:0.022
M:2.
N:0.14
GSC:1.
ISC:1.
IO:8.

ART3

[INSUL]
I:0.
Y2:0.
X:0.
ACC:0.
CRK:200.
G0:67.
K:3.3
G1:300.
LG:1.
XMAX:240.
ISR:1.
SRD:0.
ALF:0.1
BET:0.3
M:2.
N:0.333

GSC:1.
ISC:1.
IO:8.

VEN1

[INSOL]
I:0.
Y2:0.
X:0.
ACC:0.
CRK:200.
GD:67.
K:3.3
G1:300.
LG:1.
XMAX:600.
ISR:0.
SRD:1.
ALF:0.2
BET:0.04
M:0.25
N:0.333
GSC:1.
ISC:1.
IO:8.
[GLUC]
X1:0.
X2:0.
A1:0.016
B1:0.6
V:0.14
A2:0.33
[DATA]
T1:2.
G1:66.
T2:4.
G2:215.
T3:12.
G3:197.
G4:156.
T4:20.
T5:30.
G5:145.
T6:50.
G6:136.
G7:120.
T7:60.
G8:109.
I1:9.
I2:73.
I3:104.
I4:42.
I5:29.
I6:17.
I7:16.
I8:16.
[IVGTT]
T1:2.
DOS:25.